
Introduktion till Maple
Allmänt

Ett modernt datoralgebrasystem har som huvudfunktion att göra symboliska beräkningar,
i motsats till numeriska. Det kan utföra algebraiska manipulationer och förenklingar, lösa
ekvationssystem, integrera och derivera symboliskt och lösa differentialekvationer. Dessutom
kan de flesta s̊adana system ocks̊a utföra numeriska beräkningar och har kraftfulla grafiska
funktioner. Behöver man mer omfattande numeriska beräkningar använder man sig dock
hellre av Matlab eller specialskrivna numeriska program.

Maple är ett av de ledande datoralgebrasystemen. N̊agra av de starkaste konkurrenterna
är Mathematica, Macsyma och Reduce. LTH har en generell licens för Maple och syste-
met bör därför finnas tillgängligt p̊a studentdatorer. Det kan köras fr̊an de flesta vanliga
operativsystem.

Maple är ett mycket mångsidigt program, och man kan ägna mycket tid åt att utforska
dess användningsmöjligheter. Dess m̊angsidighet gör det samtidigt sv̊arare att använda än
t. ex. Matlab.

Liksom de flesta programpaket uppgraderas Maple relativt ofta. Versionerna skiljer sig
n̊agot åt, och anvisningarna nedan bör fungera bra för den version som finns installerad i
datorsalarna.

Mer information om Maple än vad som ryms i denna korta introduktion kan man finna
i det inbyggda hjälpsystemet samt i en stort antal skrifter och böcker.

Komma ig̊ang

Starta Maple genom att klicka p̊a ikonen p̊a skrivbordet eller i Start-menyn. Man g̊ar ur
Maple genom att välja Exit i File-menyn. Behöver man avbryta n̊agon process kan man
använda stoppknappen p̊a menyraden.

Hjälp

Maple har ett omfattande inbyggt hjälpsystem. Om man klickar p̊a Help i menyraden, s̊a
f̊ar man tillg̊ang till hjälpfunktioner. Dessa fungerar p̊a lite olika sätt, beroende p̊a vilken
Mapleversion man kör, men brukar vara lätta att använda.

Man kan ocks̊a f̊a hjälp genom att skriva ett sökord föreg̊anget av ett fr̊agetecken, t. ex.
?exp, ?inifcn. I slutet p̊a varje hjälptext finns exempel och hänvisningar till andra kom-
mandon som kan vara av nytta.

Fördefinierade konstanter och funktioner

Maple har n̊agra fördefinierade konstanter (se ?constant), t. ex.

I, Pi och infinity (st̊ar för i, π resp ∞)

Dessa är reserverade namn, som inte f̊ar användas till annat. Observera att Maple skiljer p̊a
stora och sm̊a bokstäver. Exempelvis m̊aste π (3.14 . . . ) skrivas Pi (stort P). Skriver man
pi f̊ar man ett π p̊a skärmen, men är detta enbart en symbol och har inget numeriskt värde.
För talet e finns ingen symbol utan man skriver exp(1) (se ?exp).

I Maple finns m̊anga fördefinierade funktioner (se ?inifcn), däribland de välkända funk-
tionerna

sin cos tan cot arcsin arccos arctan arccot

abs sqrt exp ln log log10

(Här st̊ar abs för absolutbelopp, sqrt för kvadratroten, exp för exponentialfunktionen (ex)
b̊ade ln och log för naturliga logaritmen och log10 för tiologaritmen (lg).

1



Grundläggande kommandon och aritmetiska beräkningar

I Maple-fönstret hittar du en ’prompt’ >, som betyder att Maple väntar p̊a ett kommando.
Skriv in 7+5; och tryck p̊a Enter, s̊a utförs summationen.

Om man avslutar ett Maple-kommando med ett kolon :, s̊a utförs operationen, men
resultatet kommer inte upp p̊a skärmen. Dett kan vara användbart om man inte vill fylla
skärmen med l̊anga delräkningar som man änd̊a inte är intresserad av. (I äldre versioner av
Maple var det nödvändigt att avsluta varje kommando med ett semikolon ;, men det behövs
inte längre. I de flesta exempel nedan används änd̊a detta semikolon, mest av gammal vana.)

Kontrollera att Maple kan fungera som en vanlig räknedosa. Prova i tur och ordning

2-3*5; 1/2+1/3; sqrt(4)*sqrt(3);

Observera att Maple ger exakta svar, inte närmevärden.
Närmevärden erh̊alles genom kommandot evalf;, som st̊ar för ’evaluate using floating

point arithmetic’. För att f̊a ett närmevärde för den senast utförda beräkningen (2
√

3 i v̊art
fall) kan man skriva

evalf(%);

Man kan naturligtvis ocks̊a skriva evalf(sqrt(4)*sqrt(3));. P̊a samma sätt kan resul-
tatet av den näst senaste beräkningen åberopas genom %%, etc.

En praktisk detalj i Maple är möjligheten att g̊a tillbaka till gamla kommandon p̊a
skärmen med pil-upp och pil-ner tangenterna, ändra i kommandona och utföra dem p̊a nytt.
Man kan ocks̊a använda musen för att snabbt flytta sej till olika delar av skärmen. Testa
genom att g̊a tillbaka till kommandot 1/2+1/3; och gör en annan br̊akberäkning genom att
ändra siffror.

När det gäller kommandot evalf s̊a kan man ur hjälpfunktionen, som n̊as med ?evalf,
utläsa att man genom ett andra argument kan ange med hur m̊anga siffror man vill ha svaret.
Använd detta för att f̊a ett värde p̊a π med 1000 decimaler, evalf(Pi,1000);.

Variabler

Maple kan inte bara räkna med tal utan ocks̊a med variabler och med funktioner. Detta
gör att ett Maplesystem blir större och ofta mer invecklat att programmera än ett vanligt
programmeringspr̊ak, men ocks̊a oerhört mycket mera kraftfullt.

Ge kommandot (x+1)^3;. Maple svarar med samma sak. Som svar p̊a expand(%); s̊a
utvecklar Maple uttrycket enligt binomialsatsen. Prova även med

expand((a-b)*(a+b)); expand(x*(x+1)*(x+2)*(x+3)); expand(cos(x+y));

Som bekant vill man ofta g̊a åt andra h̊allet, och faktoruppdela ett givet uttryck. Detta
görs med kommandot factor. Prova detta p̊a de polynomuttryck som du fick ovan.

Ett annat kommando för förenkling är simplify. Prova detta genom att skriva

simplify(1/(x-1)+1/(x+1));

Maple gör inte alltid det man önskar. Ett annat kommando man kan ha nytta av för att
göra omskrivningar är normal.

Tilldelningssatser

Man kan tilldela en variabel ett värde, numeriskt eller symboliskt. Tilldelningssymbolen är
liksom i Pascal och Simula :=. Prova med x:=2;. Kommandot x; ger nu variabelvärdet 2.
Prova ocks̊a med (x+1)^2;.

En variabel som f̊att ett värde beh̊aller detta tills man ger den ett annat värde eller tar
bort värdet eller g̊ar ur Maple. För att ta bort värdet använder man kommandot x:=’x’; ,
som gör att Maple tolkar bokstaven x som en variabel, betecknad x, inte som variabelns
värde. Det är lätt att glömma bort att man gett en variabel ett värde tidigare, vilket kan
leda till obegripliga resultat av räkningar. Vill man ta bort alla värden p̊a variabler och helt
och h̊allet börja om fr̊an början skriver man restart;.

Variabler kan förutom numeriska värden även ha Mapleuttryck som värden. Genom till-
delningskommandot

f:=(x+1)^3;

s̊a sätter vi f lika med (x + 1)3. (Blir svaret 27 s̊a tag bort det tidigare värdet fr̊an x och
försök igen.) Man kan sedan räkna vidare med f och skriva t. ex. f^2; och expand(f^2);.

Om man vill beräkna värdet av uttrycket f^2 för t. ex. x=2 utan att varaktigt tilldela x

ett värde s̊a kan man skriva

eval(x=2,f^2); eller subs(x=2,f^2);

Funktioner

Maple kan hantera inte bara analytiska uttryck, som vi sett exempel p̊a ovan, utan även
funktioner. Dessa kan definieras p̊a flera olika sätt. Det för v̊ara ändam̊al enklaste p̊aminner
om beteckningen

x 7→ g(x)

för en funktion g. Ge kommandot

g:= x->(x+1)^3;

Nu kan funktionsvärden beräknas p̊a det sätt man är van vid. Prova t. ex. med

g(0); , g(-1); , g(a); och g(y+z); .

Anmärkning: Man m̊aste noga h̊alla isär begreppen funktion och funktionsuttryck, där
det senare behandlades under rubriken ’Tilldelningssatser’ ovan. För säkerhets skull upprepar
vi skillnaden:

Om vi ger ett värde till F genom tilldelningen F:=exp(x)-sin(x); s̊a är F ett uttryck
med variabeln x inbyggt. Om vi i stället ger F ett värde genom F:=x->exp(x)-sin(x); s̊a är
F en funktion, där x bara används för att definiera en regel, och där man skulle kunna använda
vilken annan symbol som helst. Funktioner är mycket mer flexibla, men ibland n̊agot mer
sv̊arhanterliga än funktionsuttryck. Det finns ett enkelt sätt att göra om ett uttryck till en
funktion, nämligen genom att använda kommandot unapply. Till exempel ger kommandot
F:=unapply(exp(x)-sin(x),x); samma sak som F:=x->exp(x)-sin(x);.

Maple klarar ocks̊a funktioner av flera variabler. Funktionen h(x, y) = x2 − y3 + x2y3

definieras i Maple p̊a följande sätt:

h:= (x,y) -> x^2-y^3+x^2*y^3;

Vad ger h(0,0); och h(3,2);?

Derivation

Derivator beräknas med kommandot diff. Användningen framg̊ar ur exemplen

diff(x^3+2*x^2,x); och diff(tan(x),x);

Andraderivator beräknas enligt

diff(tan(x),x,x);

P̊a samma sätt beräknas derivator av högre ordning, där antalet x anger ordningen. För
detta finns ett kortare skrivsätt. Beräkna fjärdederivatan av tan(x) genom att skriva

diff(tan(x),x$4);.

Här betyder $4 att derivationen upprepas 4 g̊anger. Prova även att förenkla svaret med hjälp
av factor(%);.

Vi kan derivera det tidigare definierade funktionsuttrycket f och funktionen g genom att
skriva

diff(f,x); resp diff(g(x),x);

Tänk ut vad som kommer ut ur kommandona

diff(f,y); , diff(g,x); , diff(g(y),y);

och prova om det stämmer.
Anmärkning: Maple har ocks̊a en derivationsoperator D, se ?D. Pröva t. ex. D(g); och

D(sin);. Bokstaven D kan inte användas som symbol för n̊agot annat.

Man kan även lätt beräkna partiella derivator. Till exempel f̊ar man

∂h

∂x
och

∂h

∂y

för funktionen h ovan genom kommandona

diff(h(x,y),x); resp diff(h(x,y),y);

Högre derivator f̊as genom uppräkning av variabelnamnen. Pröva t. ex.

diff(h(x,y),x,y);

Integration

Med hjälp av kommandot int(f,x); kan man beräkna en primitiv funktion till funktionen
f(x) med avseende p̊a x. Integrationskonstanten f̊ar man sedan själv lägga till om det skulle
behövas. Om vi definierat ett funktionsuttryck f och en funktion g som p̊a förra sidan, s̊a
f̊as en primitiv funktion med

int(f,x); resp int(g(x),x);

Repetera nu n̊agra primitiva funktioner fr̊an envariabelanalysen t. ex.: cosx (med kom-

mandot int(cos(x),x);), et (skrivet exp(t)), och
1√

1− z2
med avseende p̊a respektive

variabel. Kvadratrotfunktionen skrivs i Maple som sqrt.
Om Maple inte kan n̊agon primitiv funktion, s̊a svarar Maple genom att ge tillbaka det

man stoppat in. Försök med int(x^x,x); s̊a f̊ar du tillbaka∫
xx dx.

Prova ocks̊a ∫ √
1− 2x2

1− x2
dx och

∫
e−x

2

dx .

Dessa exempel visar att Maple känner till fler primitiva funktioner än de vanliga elementära
funktionerna. Den första av integralerna är exempel p̊a en s̊a kallad elliptisk integral. Vissa
elliptiska integraler finns i Maple, se t. ex. ?EllipticE. I svaret p̊a den andra integralen ing̊ar
den s̊a kallade ’error function’, erf. Denna uppträder bl a inom statistiken och i samband
med värmeledning.

För att beräkna integraler, till exempel∫ 3

−2
(x+ 1)3 dx

m̊aste även gränserna anges. Detta görs i Maple genom x=-2..3, eller allmänt x=a..b för
a ≤ x ≤ b. Integralen ovan beräknas genom

int((x+1)^3,x=-2..3);

Gränserna f̊ar gärna inneh̊alla variabler, försök t. ex. med int(f,x=-y..(z+1));.
Även generaliserade integraler kan beräknas. Prova t. ex. med

int(1/(1+x^2),x=0..infinity);

Använder man en stor begynnelsebokstav (t. ex. skriver Int(1/(1+x^2),x=0..infinity);)
gör inte Maple n̊agon beräkning utan svarar med en snygg utskrift av det man skrivit in.
Detta är bra att använda om man vill kontrollera att man skrivit rätt.

Summation

Maple kan ocks̊a användas för att beräkna summor. Syntaxen för summation är naturlig,∑n
k=m ak f̊as genom Maplekommandot sum(a(k),k=m..n).
Beräkna potenssummorna

n∑
k=1

k och

n∑
k=1

k2

med hjälp av kommandona sum(k,k=1..n); och sum(k^2,k=1..n);. Förenkla med kom-

mandot simplify. Även vissa serier kan Maple beräkna. Vad blir
∞∑
k=1

1

k2
?

Gränsvärden

Gränsvärden beräknas med kommandot limit, prova

limit(sin(x)/x,x=0); limit((sqrt(1+exp(x))-1)/exp(x),x=infinity);

Taylor- och Maclaurinutvecklingar

Med kommandot taylor f̊ar man Taylor- (Maclaurin)utvecklingen till ordning 5, t. ex. ger

taylor(exp(sin(x)),x);

Maclaurinutvecklingen av esin x. Vill man istället taylorutvecka sinx kring punkten x = 1 till
ordning 9 (restterm av ordning 10) skriver man

taylor(sin(x),x=1,10);

Ekvationslösning

Maple kan användas för att lösa ekvationer, se ?solve och ?fsolve. Prova de tre varianterna

solve(x^3+x^2+x-3=0,x);

fsolve(x^3+x^2+x-3=0,x);

fsolve(x^3+x^2+x-3=0,x,complex);

och läs om dem i hjälpfunktionen. Pröva ocks̊a

solve(sin(x)=1/2,x);

I det senare fallet ger Maple endast en lösning. För att f̊a alla lösningar kan man skriva

_EnvAllSolutions := true: solve(sin(x)+cos(x)=sqrt(2),x);

Differentialekvationer

Maple kan ocks̊a lösa differentialekvationer, se ?dsolve. Pröva med att lösa

y′′ + 2y′ + 2y = e−x

med kommandot

dsolve(diff(y(x),x$2)+2*diff(y(x),x)+2*y(x)=exp(-x), y(x));

Har vi dessutom begynnelsevillkoren y(0) = y′(0) = 0 kan vi skriva

dsolve({diff(y(x),x$2)+2*diff(y(x),x)+2*y(x)=exp(-x),

y(0)=0,D(y)(0)=0}, y(x));

Grafik

Maple har omfattande grafiska möjligheter. Om du vill att Maples figurer hamnar i egna
fönster (vilket kan vara lättare att hantera) s̊a g̊ar du in under menyn File, väljer Preferences
och sen Plotting, Window. Antalet grafikfönster har en tendens att växa snabbt. Man stänger
dem man inte behöver genom att g̊a in under File och dra till Close när grafikfönstret är
uppe.

Med kommandot plot ritar man tv̊adimensionella figurer i Maple, t. ex. funktionskurvor.
Se hjälpen till kommandona plot. Prova med

plot(sin(x),x=-Pi..Pi);

Axlarna skalas automatiskt s̊a att bilden fyller grafikfönstret maximalt. Om man vill
ha samma skala p̊a b̊ada axlarna kan man g̊a in p̊a menyraden vid Projection och välja
Constrained. Detta kan man ocks̊a åstadkomma genom att direkt skriva

plot(sin(x),x=-Pi..Pi,scaling=constrained);

Vill man hellre ha en grön kurva skriver man

plot(sin(x),x=-Pi..Pi,scaling=constrained,color=green);

Man kan ocks̊a välja olika typer av axlar, linjetjocklek, linjestil antingen fr̊an menyraden
eller direkt p̊a kommadoraden.

För att rita flera kurvor i en figur använder man [ ]-parenteser (även {}-parenteser fun-
gerar). Pröva

plot([sin(x),x-x^3/6],x=-Pi..Pi);

Vad är det för speciellt med polynomet?
Man kan ocks̊a rita kurvor i parameterform. Enhetsirkeln (x, y) = (cos t, sin t), −π ≤ t ≤

π ritas med

plot([cos(t),sin(t),t=-Pi..Pi],scaling=constrained);

Observera att parameterintervallet skrivs innanför hakparentesen.
Maple har m̊anga fler grafikrutiner. För att f̊a tillg̊ang till dessa görs kommandot

with(plots);

Här är plots namnet p̊a ett s̊a kallat ’package’. Det finns ett antal s̊adana i Maple, för olika
användningar. Avslutar man med ; f̊ar man en lista över de nya kommandon man f̊ar tillg̊ang
till. Ett annat ’package’ för grafik är plottools.

Nu kan vi ocks̊a rita implicit givna kurvor, t. ex.
x2

4
+
y2

9
= 1, med kommandot

implicitplot. Ange ekvationen samt ett lämpligt intervall för x och y.

implicitplot(x^2/4+y^2/9=1,x=-3..3,y=-3..3);

Välj här Projection Constrained.
Ibland är det bra att först definiera och lagra bilden och sen titta p̊a den. Om man vill

se de b̊ada kurvorna x2/4 + y2/9 = 1 och (x, y) = (2 cos t, 3 sin 3t), 0 ≤ t ≤ 2π i samma bild
kan man göra s̊a här: L̊at A betyda bilden av den första kurvan och B bilden av den andra,
skriv

A:=implicitplot(x^2/4+y^2/9=1,x=-3..3,y=-3..3,color=red):

B:=plot([2*cos(t),3*sin(3*t),t=0..2*Pi],color=green):

Avsluta här med :, ej med ;. (Avslutar man med ; här visas ingen bild, utan skärmen fylls
av siffror.) Maple visar nu inga bilder utan lagrar bilderna A resp B. Om vi nu skriver

display([A,B]);

visar Maple b̊ada kurvorna i en bild. Välj här Projection Constrained.
Maples tredimensionella grafik har en mängd varianter, läs om dem i ?plot3d.

Funktionsytan z = xe−x
2−y2

, −2 ≤ x ≤ 2, −2 ≤ y ≤ 2 kan ritas med

plot3d(x*exp(-x^2-y^2),x=-2..2,y=-2..2);

Man kan vrida och vända p̊a bilden genom att placera musmarkören i grafikfönstret och h̊alla
vänster musknapp nere och flytta p̊a musen eller klicka pilarna bredvid vinkelbeteckningarna
ϑ och φ i menyraden. Prova med hjälp av menyer de olika möjligheterna för ’axes’, ’color’
och ’style’.

Definitionsmängden behöver ej vara en rektangel, gränserna i y kan bero p̊a x, pröva
t. ex.

plot3d(x*exp(-x^2-y^2),x=0..2,y=-x..x);

Vad är definitionsmängden här?
Vill man bara se funktionens niv̊akurvor skriver man

contourplot(x*exp(-x^2-y^2),x=-2..2,y=-2..2);

Det g̊ar ocks̊a bra att rita ytor i parameterform. Cylindern y2 + z2 = 1 längs x-axeln kan
ritas med

plot3d([s,cos(t),sin(t)],s=-2..2,t=0..2*Pi,scaling=constrained);

Observera att parameterintervallen här skrivs utanför hakparentesen. (Här skiljer sig allts̊a
det tre- och tv̊adimensionella fallen åt.)

Vill man titta p̊a tv̊a cylindrar (en längs x-axeln och en längs y-axeln) som skär varandra
kan man göra s̊a här:

A:=plot3d([s,cos(t),sin(t)],s=-2..2,t=0..2*Pi):

B:=plot3d([cos(t),s,sin(t)],s=-2..2,t=0..2*Pi,color=s*t):

display([A,B]);

Här är gjort ett färgval p̊a cylindern B, färgen p̊a cylindern A kan varieras fr̊an menyraden.
Välj här Projection Constrained. Vrid och vänd p̊a figuren. Pröva olika möjligheter av ’color’
och ’style’. Vill man särskilt studera skärningen mellan cylindrarna (y2+z2 = 1 resp x2+z2 =
1) dvs den kropp som ligger ’innanför’ b̊ada cylindrarna skriver man

plot3d({-min(sqrt(1-x^2),sqrt(1-y^2)),min(sqrt(1-x^2),sqrt(1-y^2))},

x=-1..1,y=-1..1);

Pröva även olika Light Scheme under Color.
För att rita cylindrar, och andra ytor med rotationssymmetri kring z-axeln, snyggt är

det bra att använda cylinderplot. Exempelvis f̊as konen z =
√
x2 + y2, x2 + y2 ≤ 4 av

cylinderplot(z,t=0..2*Pi,z=0..2);

Här anges avst̊andet fr̊an z-axeln (’radien’) r =
√
x2 + y2 som funktion av z. Den första

variabeln anger vinkeln till x-axeln. Paraboloiden z = x2 + y2, (r =
√
x2 + y2 =

√
z) f̊as av

cylinderplot(sqrt(z),t=0..2*Pi,z=0..2);

Samma sak erh̊alls med

plot3d(sqrt(z),t=0..2*Pi,z=0..2,coords=cylindrical);

För mera komplicerade ytor se ?cylinderplot.
En sfär kan även ritas med kommandot

sphereplot(1,t=0..2*Pi,s=0..Pi,scaling=constrained);

alternativt

plot3d(1,t=0..2*Pi,s=0..Pi,coords=spherical,scaling=constrained);

Här är den första varibeln den vinkel som kallas ϕ i flerdimboken och den andra variabeln
θ, se även ?sphereplot och ?coords.

Implicit givna ytor kan ritas med implicitplot3d, pröva t. ex.

implicitplot3d(x^2+y^2-z^2=1,x=-2..2,y=-2..2,z=-2..2);

(Denna yta kan ritas snyggare med cylinderplot.) Prova ocks̊a

implicitplot3d(x^2+4*y^2+9*z^2=1,x=-1..1,y=-1/2..1/2,

z=-1/3..1/3,scaling=constrained);

Fler roliga exempel p̊a funktionsytor finns att hämta i övningshäftet i flervariabelanalys.
Kurvor i rummet ritas med spacecurve, prova

spacecurve([t*cos(t),t*sin(t),t],t=0..3*Pi):

Här skall parameterintervallet skrivas utanför hakparentesen.
Vill man se kurvan tydligare kan man lägga in den p̊a en kon (jfr flerdimboken sid 24)

skriv

A:=spacecurve([t*cos(t),t*sin(t),t],t=0..3*Pi,color=red,thickness=4):

B:=cylinderplot(z,t=0..2*Pi,z=0..10):

display([A,B]);

Spara och skriva ut

Man kan spara all text i Maple-fönstret genom att klicka p̊a File och Save as samt ange
ett filnamn. Texten sparas d̊a i en fil med namnet filnamn.mws. När du senare vill ta fram
texten igen klickar du p̊a File och Open och anger hela filnamnet, även .mws. Det g̊ar ocks̊a
att samla alla kommandon i en scriptfil med ett enkelt namn, t. ex. prov. För att köra scriptet
skriver du i Maple-fönstret read prov;. I vissa versioner används filändelsen .mw i stället
för .mws.

Om du vill ta ut en bild p̊a skrivare, kan du spara den i Post-Script format genom att
klicka p̊a File och Print samt ange filnamn.ps. Alternativt kan man skicka bilden direkt till
skrivaren.

